<jupyter_start><jupyter_text>--- Day 17: Trick Shot --- [![Binder](https://mybinder.org/badge_logo.svg)](https://mybinder.org/v2/gh/oddrationale/AdventOfCode2021FSharp/main?urlpath=lab%2Ftree%2FDay17.ipynb) You finally decode the Elves' message. HI, the message says. You continue searching for the sleigh keys.Ahead of you is what appears to be a large ocean trench. Could the keys have fallen into it? You'd better send a probe to investigate.The probe launcher on your submarine can fire the probe with any integer velocity in the x (forward) and y (upward, or downward if negative) directions. For example, an initial x,y velocity like 0,10 would fire the probe straight up, while an initial velocity like 10,-1 would fire the probe forward at a slight downward angle.The probe's x,y position starts at 0,0. Then, it will follow some trajectory by moving in steps. On each step, these changes occur in the following order:The probe's x position increases by its x velocity.The probe's y position increases by its y velocity.Due to drag, the probe's x velocity changes by 1 toward the value 0; that is, it decreases by 1 if it is greater than 0, increases by 1 if it is less than 0, or does not change if it is already 0.Due to gravity, the probe's y velocity decreases by 1.For the probe to successfully make it into the trench, the probe must be on some trajectory that causes it to be within a target area after any step. The submarine computer has already calculated this target area (your puzzle input). For example:target area: x=20..30, y=-10..-5This target area means that you need to find initial x,y velocity values such that after any step, the probe's x position is at least 20 and at most 30, and the probe's y position is at least -10 and at most -5.Given this target area, one initial velocity that causes the probe to be within the target area after any step is 7,2:.........................................................................................S............................................................................................................................................................................TTTTTTTTTTT....................TTTTTTTTTTT....................TTTTTTTTTT....................TTTTTTTTTTT....................TTTTTTTTTTT....................TTTTTTTTTTTIn this diagram, S is the probe's initial position, 0,0. The x coordinate increases to the right, and the y coordinate increases upward. In the bottom right, positions that are within the target area are shown as T. After each step (until the target area is reached), the position of the probe is marked with . (The bottom-right  is both a position the probe reaches and a position in the target area.)Another initial velocity that causes the probe to be within the target area after any step is 6,3:....................................................................................................................................................................................S............................................................................................................................................................................TTTTTTTTTTT....................TTTTTTTTTTT....................TTTTTTTTTTT....................TTTTTTTTTTT....................TTTTTTTTTT....................TTTTTTTTTTTAnother one is 9,0:S...........................................................................................................................................................................TTTTTTTTTTT....................TTTTTTTTTT....................TTTTTTTTTTT....................TTTTTTTTTTT....................TTTTTTTTTTT....................TTTTTTTTTTTOne initial velocity that doesn't cause the probe to be within the target area after any step is 17,-4:S.............................................................................................................................................................................................................................................................................................................................................TTTTTTTTTTT....................................................TTTTTTTTTTT....................................................TTTTTTTTTTT....................................................TTTTTTTTTTT....................................................TTTTTTTTTTT...................................................TTTTTTTTTTT..................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................................The probe appears to pass through the target area, but is never within it after any step. Instead, it continues down and to the right - only the first few steps are shown.If you're going to fire a highly scientific probe out of a super cool probe launcher, you might as well do it with style. How high can you make the probe go while still reaching the target area?In the above example, using an initial velocity of 6,9 is the best you can do, causing the probe to reach a maximum y position of 45. (Any higher initial y velocity causes the probe to overshoot the target area entirely.)Find the initial velocity that causes the probe to reach the highest y position and still eventually be within the target area after any step. What is the highest y position it reaches on this trajectory?<jupyter_code>let input = File.ReadAllText @"input/17.txt"
type Position = {
    X: int
    Y: int
}

type Velocity = {
    dX: int
    dY: int
}
let parse (input: string) = 
    let split1 = input.Split(": ").[1].Split(", ")
    let splitX = split1.[0].Split("=").[1].Split("..")
    let splitY = split1.[1].Split("=").[1].Split("..")
    {
        X = int splitX.[0]
        Y = int splitY.[0]
    },
    {
        X = int splitX.[1]
        Y = int splitY.[1]
    }
let rec steps (position, velocity) = 
    let drag = function
        | dx when dx > 0 -> dx - 1
        | dx when dx < 0 -> dx + 1
        | _ -> 0

    let step (position, velocity) =
        {
            X = position.X + velocity.dX
            Y = position.Y + velocity.dY
        },
        {
            dX = drag velocity.dX
            dY = velocity.dY - 1
        }

    seq {
        yield position, velocity
        yield! steps (step (position, velocity))
    }
let withinTargets steps (targetStart, targetEnd) = 
    steps
    |> Seq.map (fun (p, _) -> p)
    |> Seq.takeWhile (fun p -> 
        p.X <= targetEnd.X && 
        p.Y >= targetStart.Y)
    |> Seq.exists (fun p -> 
        targetStart.X <= p.X && p.X <= targetEnd.X && 
        targetStart.Y <= p.Y && p.Y <= targetEnd.Y)
let calcHeight velocity = 
    (velocity.dY*(velocity.dY + 1))/2
#!time
let (targetStart, targetEnd) = input |> parse

let hits = 
    ([1..targetEnd.X], [targetStart.Y..abs targetStart.Y])
    ||> Seq.allPairs 
    |> Seq.map (fun (x, y) -> {dX = x; dY = y})
    |> Seq.filter (fun v -> withinTargets (steps ({X = 0; Y = 0}, v)) (targetStart, targetEnd))
    |> Seq.cache

hits
|> Seq.map calcHeight
|> Seq.max<jupyter_output><empty_output><jupyter_text>--- Part Two --- Maybe a fancy trick shot isn't the best idea; after all, you only have one probe, so you had better not miss.To get the best idea of what your options are for launching the probe, you need to find every initial velocity that causes the probe to eventually be within the target area after any step.In the above example, there are 112 different initial velocity values that meet these criteria:23,-10  25,-9   27,-5   29,-6   22,-6   21,-7   9,0     27,-7   24,-525,-7   26,-6   25,-5   6,8     11,-2   20,-5   29,-10  6,3     28,-78,0     30,-6   29,-8   20,-10  6,7     6,4     6,1     14,-4   21,-626,-10  7,-1    7,7     8,-1    21,-9   6,2     20,-7   30,-10  14,-320,-8   13,-2   7,3     28,-8   29,-9   15,-3   22,-5   26,-8   25,-825,-6   15,-4   9,-2    15,-2   12,-2   28,-9   12,-3   24,-6   23,-725,-10  7,8     11,-3   26,-7   7,1     23,-9   6,0     22,-10  27,-68,1     22,-8   13,-4   7,6     28,-6   11,-4   12,-4   26,-9   7,424,-10  23,-8   30,-8   7,0     9,-1    10,-1   26,-5   22,-9   6,57,5     23,-6   28,-10  10,-2   11,-1   20,-9   14,-2   29,-7   13,-323,-5   24,-8   27,-9   30,-7   28,-5   21,-10  7,9     6,6     21,-527,-10  7,2     30,-9   21,-8   22,-7   24,-9   20,-6   6,9     29,-58,-2    27,-8   30,-5   24,-7How many distinct initial velocity values cause the probe to be within the target area after any step?<jupyter_code>#!time
hits
|> Seq.length<jupyter_output><empty_output>
{'fraction_non_alphanumeric': 0.3420479302832244, 'fraction_numerical': 0.043899782135076254, 'mean_word_length': 1.179724596391263, 'pattern_counts': {'":': 1, '<': 15, '<?xml version=': 0, '>': 30, 'https://': 2, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 24, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'chain_length': '2', 'dir': 'jupyter-structured-clean-dedup', 'id': '500945', 'license': 'MIT', 'n_tokens_mistral': 2358, 'n_tokens_neox': 2137, 'n_words': 903, 'path': 'Day17.ipynb', 'repo_name': 'oddrationale/AdventOfCode2021FSharp', 'size': '9180'}